
1  Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-EUTRA-B2013-03_D13k49’.

 Corrections

3.1     Change 1 – Correction to record “a_GERAN_Receive_SNDCPData” 
	Record name
	a_GERAN_Receive_SNDCPData

	Reason for change
	The TTCN implementation in altstep a_GERAN_Receive_SNDCPData does not ensure that IP loopback data has actually been received on the G_SN port. It simply assumes is that the data has been received if the UE sends a PACKET CONTROL ACKNOWLEDGEMENT. However, this could result in the TC implementation passing non-conformant UEs.

	Summary of change
	Introduced a global variable and associated functions to ensure that the IP loopback data had indeed been received at the G_SN port.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before change:
	/* @desc      To receive Uplink Data as SNDCP from the UE

   *            PLEASE NOTE: This is the 2nd function of 2 and MUST be preceded by f_GERAN_Trigger_UplinkData

   * @param     p_CellId

   * @param     p_Msg               (default value: ?)

   * @status    APPROVED (LTE_IRAT)

   */
altstep a_GERAN_Receive_SNDCPData(GERAN_CellId_Type p_CellId,
                                    template (present) N_PDU p_Msg := ?) runs on GERAN_PTC
  {
    timer t_WaitForMoreData := 0.1;
    // Receive the message
    [] G_SN.receive (car_G_SN_UnitData_IND(?, ?, p_Msg))
      { // ok, received another data block, so carry on
        repeat;
      }
    [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)) // RACE CONDITION
      {
        t_WaitForMoreData.start;
        alt {
        [] G_SN.receive (car_G_SN_UnitData_IND(?, ?, p_Msg))
          {
            t_WaitForMoreData.stop;
          }
        [] t_WaitForMoreData.timeout {}

      }
    }
  }

  


After change:

	/* @desc      To receive Uplink Data as SNDCP from the UE
   *            PLEASE NOTE: This is the 2nd function of 2 and MUST be preceded by f_GERAN_Trigger_UplinkData
   * @param     p_CellId
   * @param     p_Msg               (default value: ?)
   * @status    APPROVED (LTE_IRAT)
   */

  altstep a_GERAN_Receive_SNDCPData(GERAN_CellId_Type p_CellId,
                                    template (present) N_PDU p_Msg := ?) runs on GERAN_PTC
  {
    timer t_WaitForMoreData := 0.1;
    // Receive the message
    [] G_SN.receive (car_G_SN_UnitData_IND(?, ?, p_Msg))
      { // ok, received another data block, so carry on
        f_GERAN_SNDCPIPDataRcvd_Set();
        repeat;
      }
    [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)) // RACE CONDITION
      {
        t_WaitForMoreData.start;
        alt {
        [] G_SN.receive (car_G_SN_UnitData_IND(?, ?, p_Msg))
          {
            f_GERAN_SNDCPIPDataRcvd_Set();
            t_WaitForMoreData.stop;
          }
        [] t_WaitForMoreData.timeout 
          {
            if (not f_GERAN_SNDCPIPDataRcvd_Get())
            {
              t_WaitForMoreData.start;
              repeat;  
            }     
          }
        }
      }
  }




3.2     Change 2 – Correction to function “f_GERAN_Init”
	Function name
	f_GERAN_Init

	Reason for change
	See change 1 above.

	Summary of change
	Added and initialized the new variable vc_GERAN_Global.SNDCPIPDataRcvd.

	TTCN module
	GERAN_CellInfo.ttcn

	MCC160 Comment
	


Before change:

	function f_GERAN_Init(IRAT_ConfiguredPTCs_Type p_ConfiguredPTCs := EUTRA_GERAN) runs on GERAN_PTC
  { /* @sic R5-113805, R5-113696, R5-113734 sic@ */
    /* f_GERAN_Init is used as dummy function (without parameters)  for the GERAN PTC when the GERAN PTC is needed just to get the GERAN system information */
    var IRAT_CoOrd_SysInfo_List_Type v_IR_SysInfoList;
    vc_GERAN_Global.Ctrl.Default := null;
    vc_GERAN_Global.Ctrl.TestBody := false;
    vc_GERAN_Global.Security := f_GERAN_Security_Init();
    vc_GERAN_Global.CellArray := f_GERAN_CellArray_Init();
    vc_GERAN_Global.MsgInDefault := f_GERAN_MsgInDefault_Init ();
    vc_GERAN_Global.SNDCPConfigured := false; // @sic R5s130777 sic@
    // Now initialize sys info, together with EUTRA inter RAT parameters, for all cells
    select (p_ConfiguredPTCs)
    {
      case (EUTRA_UTRAN_GERAN) {
        // update sys info with UTRAN inter RAT parameters for all cells
        vc_GERAN_Global.CellArray := f_GERAN_SystemInformation_InterRAT(vc_GERAN_Global.CellArray, f_IRAT_InitialCoOrd_2Ports(EUTRA, UTRAN, f_GERAN_IR_SysInfoInit (vc_GERAN_Global.CellArray)));
      }
      case (EUTRA_GERAN) {
        v_IR_SysInfoList[0] := f_IRAT_InitialCoOrd(EUTRA, f_GERAN_IR_SysInfoInit (vc_GERAN_Global.CellArray));
        vc_GERAN_Global.CellArray := f_GERAN_SystemInformation_InterRAT(vc_GERAN_Global.CellArray, v_IR_SysInfoList);
      }
      case else {
        FatalError (__FILE__, __LINE__, "wrong PTC configuration specified in test case");
      }
    }
    f_GERAN_ActivateDefault(StandardDefault);
  }




After change:

	function f_GERAN_Init(IRAT_ConfiguredPTCs_Type p_ConfiguredPTCs := EUTRA_GERAN) runs on GERAN_PTC
  { /* @sic R5-113805, R5-113696, R5-113734 sic@ */
    /* f_GERAN_Init is used as dummy function (without parameters)  for the GERAN PTC when the GERAN PTC is needed just to get the GERAN system information */
    var IRAT_CoOrd_SysInfo_List_Type v_IR_SysInfoList;
    vc_GERAN_Global.Ctrl.Default := null;
    vc_GERAN_Global.Ctrl.TestBody := false;
    vc_GERAN_Global.Security := f_GERAN_Security_Init();
    vc_GERAN_Global.CellArray := f_GERAN_CellArray_Init();
    vc_GERAN_Global.MsgInDefault := f_GERAN_MsgInDefault_Init ();
    vc_GERAN_Global.SNDCPConfigured := false; // @sic R5s130777 sic@
    vc_GERAN_Global.SNDCPIPDataRcvd := false;
    // Now initialize sys info, together with EUTRA inter RAT parameters, for all cells
    select (p_ConfiguredPTCs)
    {
      case (EUTRA_UTRAN_GERAN) {
        // update sys info with UTRAN inter RAT parameters for all cells
        vc_GERAN_Global.CellArray := f_GERAN_SystemInformation_InterRAT(vc_GERAN_Global.CellArray, f_IRAT_InitialCoOrd_2Ports(EUTRA, UTRAN, f_GERAN_IR_SysInfoInit (vc_GERAN_Global.CellArray)));
      }
      case (EUTRA_GERAN) {
        v_IR_SysInfoList[0] := f_IRAT_InitialCoOrd(EUTRA, f_GERAN_IR_SysInfoInit (vc_GERAN_Global.CellArray));
        vc_GERAN_Global.CellArray := f_GERAN_SystemInformation_InterRAT(vc_GERAN_Global.CellArray, v_IR_SysInfoList);
      }
      case else {
        FatalError (__FILE__, __LINE__, "wrong PTC configuration specified in test case");
      }
    }
    f_GERAN_ActivateDefault(StandardDefault);
  }




3.3     Change 3 – Correction to record “vc_GERAN_Global”
	Record name
	vc_GERAN_Global

	Reason for change
	See change 1 above.

	Summary of change
	Extended “GERAN_Global_Type” to include boolean type SNDCPIPDataRcvd and defined a new functiuon to set its value.

	TTCN module
	GERAN_Component.ttcn

	MCC160 Comment
	


Before change:

	type record GERAN_Global_Type { /* @status    APPROVED (IMS_IRAT, LTE_A, LTE_IRAT, UTRAN) */
    GERAN_Ctrl_Type             Ctrl,
    GERAN_SecurityInfo_Type     Security,
    GERAN_CellArray_Type        CellArray,
    GERAN_MsgReceivedInDefault_Type   MsgInDefault,
    boolean                     SNDCPConfigured,   // @sic R5s130777 sic@
  };



After change:

	type record GERAN_Global_Type { /* @status    APPROVED (IMS_IRAT, LTE_A, LTE_IRAT, UTRAN) */
    GERAN_Ctrl_Type             Ctrl,
    GERAN_SecurityInfo_Type     Security,
    GERAN_CellArray_Type        CellArray,
    GERAN_MsgReceivedInDefault_Type   MsgInDefault,
    boolean                     SNDCPConfigured,   // @sic R5s130777 sic@
    boolean                     SNDCPIPDataRcvd   
  };




New Functions:

	  function f_GERAN_SNDCPIPDataRcvd_Set() runs on GERAN_PTC
  {
    vc_GERAN_Global.SNDCPIPDataRcvd := true;
  }
  function f_GERAN_SNDCPIPDataRcvd_Get() runs on GERAN_PTC return boolean
  {
    return vc_GERAN_Global.SNDCPIPDataRcvd;
  }



